# Power plants optimisation

Applied matemathical modelling
to natural resources management

Author: Valentino Mascherini
Matrikelnummer: 12235558

20-12-22

## 1 Research problem description

The capital of Germany, Berlin, is surrounded by many different power plants, which run on different fuels or renewable energies, and supply the city. Those shall be managed in a way that the cost is minimal, by finding the optimal share of power generated by each fuel-powered power plant group, at for each month and period of the day $POW_{f,m,p}$ (or $\theta_{f,m,p}$ in the equations) in MWh. This research considers the power plants in a range of 65 km around Berlin. The fuel types ($f$) are biomass, coal, gas, oil, solar, waste and wind, and are characterised by total generation capacity per month in different periods of the day ($SUPPLYDAY_{f,m}$ and $SUPPLYNIGHT_{f,m}$) in MWh. They also have unique emission potentials for different greenhouse gasses ($CO_2POT_f$,$CH_4POT_f$,$N_2OPOT_f$) in kg/MWh and economic parameters such are fuel cost ($FC_f$) and operational costs ($OC_f$) both in $euro$/MWh. The demanded capacity for Berlin is stored in the monthly demand by day and night ($POWUSE_{m,p}$) in MWh.

## 2 Research objective

The objective of this research is to build a linear program for optimal power plant management that minimises total costs while subjected to power constraint for each month and period of the day, in order to keep the supply at least equal to the demand. Environmental constraints in terms of emission limits will also be placed on a second model, to simulate the implementation of a green city policy. The results will include the total costs, the level of power that each power plant fuel group will have to produce, and the consequent emissions.

## 3 Data and model description

### 3.1 Data

The data regarding the power plants location and generation capacity has been retrieved from the *Global Power Plant Database*. We can see in figure 1 that the majority of those are solar powered, but as figure 2 shows, most of the generation power comes from non-renewable sources. Table 1 and 2 show respectively $SUPPLYDAY_{f,m}$ and $SUPPLYNIGHT_{f,m}$, that is the maximum supplied capacity, and come from an *Eurostat* report. Table 3 contains the relevant parameter to this research, such data was found in a *EU* paper.




Figure 1: Power plants in a 65 km range from Berlin. The colours represent the different types of fuels or renewable energies employed.




Figure 2: Maximum generation capacity by power plant type in MWh near the city of Berlin.

Table 1: Table showing the sum of the maximum capacity generated during the day by different types of power plants.

| fuel | m1 | m2 | m3 | m4 | m5 | m6 | m7 | m8 | m9 | m10 | m11 | m12 |
|---|---|---|---|---|---|---|---|---|---|---|---|---|
| Biomass | 69 | 69 | 69 | 69 | 69 | 69 | 69 | 69 | 69 | 69 | 69 | 69 |
| Coal | 941 | 941 | 941 | 941 | 941 | 941 | 941 | 941 | 941 | 941 | 941 | 941 |
| Gas | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 |
| Oil | 879 | 879 | 879 | 879 | 879 | 879 | 879 | 879 | 879 | 879 | 879 | 879 |
| Solar | 301 | 469 | 469 | 536 | 402 | 872 | 704 | 402 | 603 | 637 | 436 | 368 |
| Waste | 33 | 33 | 33 | 33 | 33 | 33 | 33 | 33 | 33 | 33 | 33 | 33 |
| Wind | 144 | 129 | 153 | 169 | 171 | 121 | 134 | 140 | 121 | 155 | 155 | 152 |

Table 2: Table showing the sum of the maximum capacity generated during the night by different types of power plants.

| fuel | m1 | m2 | m3 | m4 | m5 | m6 | m7 | m8 | m9 | m10 | m11 | m12 |
|---|---|---|---|---|---|---|---|---|---|---|---|---|
| Biomass | 69 | 69 | 69 | 69 | 69 | 69 | 69 | 69 | 69 | 69 | 69 | 69 |
| Coal | 941 | 941 | 941 | 941 | 941 | 941 | 941 | 941 | 941 | 941 | 941 | 941 |
| Gas | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 | 2036 |
| Oil | 879 | 879 | 879 | 879 | 879 | 879 | 879 | 879 | 879 | 879 | 879 | 879 |
| Solar | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 0 |
| Waste | 33 | 33 | 33 | 33 | 33 | 33 | 33 | 33 | 33 | 33 | 33 | 33 |
| Wind | 7 | 6 | 7 | 8 | 8 | 6 | 6 | 7 | 6 | 7 | 7 | 7 |

Table 3: Table showing the economical and emission paramters of the different fuel types. The emission potential for CO2, co2pot, is in kg/MWh, while the other two GHG are in kg/GWh. The fuel and operational costs are both in euro/MWh

| fuel | co2pot | ch4pot | n2opot | fuelcost | opcost |
|---|---|---|---|---|---|
| Biomass | 154 | 0.000000 | 0.00000 | 0.00000 | 3.96 |
| Coal | 110 | 5.500000 | 2.00000 | 6.48000 | 3.20 |
| Gas | 65 | 3.000000 | 2.00000 | 130.00000 | 2.56 |
| Oil | 85 | 8.000000 | 1.00000 | 55.14016 | 2.50 |
| Solar | 0 | 0.000000 | 0.00000 | 0.00000 | 0.10 |
| Waste | 382 | 1.088256 | 1.93433 | 0.00000 | 1.00 |
| Wind | 0 | 0.000000 | 0.00000 | 0.00000 | 0.03 |

## 3.2 Model Description

We have a free variable expressing the total cost (TotCost), then three positive ones: $POW_{f,m,p}$, $CO_2TOT$, $CH_4TOT$ and $N_2OTOT$. The last three are calculated in the base model and the used a a reference.

$$\sum_{f,m,p} (CO2POT_f \ \theta_{f,m,p}) = CO_2TOT \tag{1}$$

$POW_{f,m,p}$ is the independent variable that influences the objective function, and figures as $\theta_{f,m,p}$ in the equations.

The model uses an objective function, three power constraints and three emission constraints. The objective function describes the total costs (TotCost) and is to be minimised. It is so defined:

$$\sum_{f,m,p} (FUELCOST_f \ \theta_{f,m,p} + OPCOST_f \ \theta_{f,m,p}) = TOTCOST \tag{2}$$

with $\theta_{f,m,p}$ being a positive variable describing the fraction of power in MWh that each power plant type contributes with, in the model called $POW_{f,m,p}$, and the costs expressed in euro/MWh.

The first constraint ensures that the total power generated is at greater or equal to the demand, in a certain month in the different periods of the day:

$$\sum_{m,p} \theta_{f,m,p} \geq POWUSE_{m,p} \ \ \forall \ \ m, p \tag{3}$$

The constraints regard the supplied capacity for day and night, and are so defined:

$$\theta_{f,m,p} \leq SUPPLYDAY_{f,m} \ \ \forall \ \ f, m \tag{4}$$

This makes sure that the power employed by each power plant type doesn't exceed its maximum generation capacity during th day. An analogue equation is employed during the night.

$CO_2$ emission constraint:

$$\sum_{f,m,p} (CO_2POT_f \ \theta_{f,m,p}) \leq REFCO_{2_f} \ EMISR \ \ \forall \ \ f \tag{5}$$

The sum of emissions for a certain GHG shouldn't exceed the reference emissions in the base scenario multiplied by the reduction coefficient, which in our model is just 0.9.

$$\sum_{f,m,p} (CH_4POT_f \ \theta_{f,m,p} \ 10^{-3}) \leq REFCH_{4_f} \ EMISR \ \ \forall \ \ f \tag{6}$$

For the $CH_4$ and $N_2O$ emissions, a factor of $10^{-3}$ is multiplied to change the unit from kg/GWh to kg/MWh.

Table 4: Total costs for each scenario in euro.

| base | Env |
|---|---|
| 374336 | 1811784 |

# 4 Results

Here we visualise the results of the base and emission constrained models.




Figure 3: Contribution of each generator type for each month in MWh/month in base scenario.




Figure 4: Contribution of each generator type for each month in MWh/month in emission reduction scenario.

Table 5: Total emissions for each scenario in kg.

| mod | CO2 | CH4 | N2O |
|---|---|---|---|
| base | 3025825 | 135.3144 | 44.74717 |
| Env | 2465824 | 121.7700 | 40.23000 |




Figure 5: Total N2O emission generation from different plants by fuel and month in kg. We can see how the coal produces the majority of the N2O.




Figure 6: Total N2O emission generation from different plants by fuel and month in kg. We can see how the gas produces the majority of the N2O.

# 5 Conclusions

As we see from figure 3, in the base scenario the coal is the most used fuel source chosen by the model, and the reason for that is clear by looking at table 3. Since the base scenario aims at minimising the costs, it firstly selects the lower cost sources, which are the renewables, since they have no fuel cost. Then, it proceeds with a merit order oriented strategy to pick the less expensive one between coal, gas and oil. Since the total supplied capacity is greater than the total demanded capacity, gas doesn't even have to be employed in the base scenario. In certain months, even oil is not used. We can see how solar power plants are affected by the weather, and in the rainiest months they do not perform as well as in the driest ones. Wind is also affected by seasonality, although not particularly. Figure 4 shows how the configuration changes after applying the environmental constraints. Gas here is the main source of fuel, since it's the cleanest, and coal has almost disappeared.

Table 4 shows the differences in cost that reducing every GHG emission by 10% entails. It is quite costly, especially reducing CO2 emissions, since switching from coal to gas is the most costly transition. We can see how much the emissions where abated in kg/year by looking at table 5. Figure 5 and 6 show for example how $N_2O$ emissions changed by type of fuel.

With this program, a technician can advise policy makers in regard to emission abatement policies and find the most suitable gradient along the Pareto optimal curve (Figure 7) between costs and emission reduction by trying out different emission reduction scenarios.




Figure 7: Example of a non-dominant Pareto front between costs and emission reduction.

# 6 References

Very rough idea for the model from:

https://www.scirp.org/journal/paperinformation.aspx?paperid=90326

Power plant data:

http://datasets.wri.org/dataset/globalpowerplantdatabase

Emission parameters:

https://op.europa.eu/en/publication-detail/-/publication/221658dd-9556-4591-86ea-51544346a8f7 https://www.epa.gov/ghgreporting/data-sets

Berlin energy use data:

https://www.stromnetz.berlin/uber-uns/zahlen-daten-fakten

Cost parameters:

https://energy.ec.europa.eu/system/files/2020-10/final_report_levelised_costs_0.pdf https://www.statista.com

Weather data:

https://meteostat.net

# 7 Appendix with GAMS code

## 7.1 Import data

```
sets
    time of the day
    /day, night/

    mperiod
    /M1*M12/

    fuel of power plant
    /biomass, coal, gas, oil, solar, waste, wind/

    scen incremental scenario steps
    /S1*S10/

    item
    /Capacity generated by the type of power plant MWh
    Distance from the city in m
    Renewable if the fuel is renewable
    CO2Pot emission potential of CO2 in kg per MWh
    CH4Pot emission potential of CH4 in kg per GWh
    N2OPot emission potential of N2O in kg per GWh
    Efficiency in generating power in %
    FuelCost cost of fuel to generate 1 MWh in euro per MWh
    OpCost operational cost to generate 1 MWh in euro per MWh
    /

;



Parameter PowUse(time,mperiod) power needed for the different months
$call GDXXRW .\Data_pp1.xlsx output= .\Data_pp1.gdx par=PowUse rng=monthly_pwr!a1:m3 rdim=1 cdim=1
$GDXIN .\Data_pp1.gdx
$LOAD PowUse
$GDXIN
  ;

Parameter p_data(fuel, item)
$call GDXXRW .\Data_pp1.xlsx output= .\Data_pp1.gdx par=p_data rng=plants_65!a1:n9 rdim=1 cdim=1
$GDXIN .\Data_pp1.gdx
$LOAD p_data
$GDXIN
  ;

Parameter supply_day(fuel, mperiod)
$call GDXXRW .\Data_pp1.xlsx output= .\Data_pp1.gdx par=supply_day rng=cap_month_d!a1:m8 rdim=1 cdim=1
$GDXIN .\Data_pp1.gdx
$LOAD supply_day
$GDXIN
  ;
```

```
Parameter supply_night(fuel, mperiod)
$call GDXXRW .\Data_pp1.xlsx output= .\Data_pp1.gdx par=supply_night rng=cap_month_n!a1:m8 rdim=1 cdim=
$GDXIN .\Data_pp1.gdx
$LOAD supply_night
$GDXIN
  ;

display
        PowUse, p_data, supply_day, supply_night;
```

### 7.2 Models

```
Parameters
            report(*,fuel, time, mperiod) universal set

            sumReport(*,*)

            emisReport(*,*, time, mperiod, fuel)

            costReport(*)
            ;

Scalar
        RefCO2 CO2 emitted without restrictions
        /3025824.7/

        RefCH4
        /135.3/

        RefN2O
        /44.7/

        EmisRed the percentage to which the emissions will be brought down to
        /0.9/
        ;

free variable
            TotCost total cost to produce energy in euro
;

positive variable
            POW(fuel, time, mperiod) power generated by each plant type each time step
            CO2Tot in kg
            CH4Tot in kg
            N2OTot in kg


;

equation
        objf minimise cost
```

```
        PowConst(time,mperiod) minimum total power needed constraint
        PowFuelDay(fuel,time, mperiod) capacity production of every plant type
        PowFuelNight(fuel,time, mperiod)
        CO2Const emission tot constraint
        CH4Const emission constraint
        N2OConst emission constraint

;

* calculate tot cost and emissions of current configuration of Berlin

objf.. TotCost
        =E=
        Sum((fuel, time, mperiod),
        p_data(fuel, "FuelCost") *
        POW(fuel, time, mperiod) +
        p_data(fuel, "OpCost") *
        POW(fuel, time, mperiod)
        );

* sum of all generated power greater than needed power

PowConst(time,mperiod)..

        Sum(fuel,
        POW(fuel, time, mperiod))
        =G=
        PowUse(time,mperiod);

* distribute total power btwn different power plants according to power constraint & min cost

PowFuelDay(fuel,time, mperiod)..
        POW(fuel, "day", mperiod)
        =L=
        supply_day(fuel, mperiod);

PowFuelNight(fuel,time, mperiod)..
        POW(fuel, "night", mperiod)
        =L=
        supply_night(fuel, mperiod);

* constrain emissions lower than normal emissions % with % as EmisRed

CO2Const..
        Sum((fuel,time, mperiod),
        p_data(fuel, "CO2Pot") *
        POW(fuel, time, mperiod)
        )
        =L=
        RefCO2 * EmisRed
;
```

```
CH4Const..
        Sum((fuel,time, mperiod),
        p_data(fuel, "CH4Pot") *
        POW(fuel, time, mperiod)/1000
        )
        =L=
        RefCH4 * EmisRed
;

N2OConst..
        Sum((fuel,time, mperiod),
        p_data(fuel, "N2OPot") *
        POW(fuel, time, mperiod)/1000
        )
        =L=
        RefN2O * EmisRed
;

* base

model plants_base /objf, PowConst, PowFuelDay, PowFuelNight/;

solve plants_base using LP min TotCost;

report("base",fuel, time, mperiod) = POW.L(fuel,time, mperiod);

* emission sum

sumReport("base","CO2") = Sum((fuel,time, mperiod),p_data(fuel, "CO2Pot") *POW.L(fuel, time, mperiod) )

sumReport("base","CH4") = Sum((fuel,time, mperiod),p_data(fuel, "CH4Pot") *POW.L(fuel, time, mperiod) /

sumReport("base","N2O") = Sum((fuel,time, mperiod),p_data(fuel, "N2OPot") *POW.L(fuel, time, mperiod) /

* emission per day and fuel

emisReport("base","CO2", time, mperiod, fuel) = (p_data(fuel, "CO2Pot") *POW.L(fuel, time, mperiod) );

emisReport("base","CH4", time, mperiod, fuel) = (p_data(fuel, "CH4Pot") *POW.L(fuel, time, mperiod) / 1

emisReport("base","N2O", time, mperiod, fuel) = (p_data(fuel, "N2OPot") *POW.L(fuel, time, mperiod) /10


costReport("base") = TotCost.L;

* env

model plants_env /objf, PowConst, PowFuelDay, PowFuelNight, CO2Const, CH4Const, N2OConst/;

solve plants_env using LP min TotCost;

report("Env",fuel, time, mperiod) = POW.L(fuel,time, mperiod);
```

```
sumReport("Env","CO2") = Sum((fuel,time, mperiod),p_data(fuel, "CO2Pot") *POW.L(fuel, time, mperiod));

sumReport("Env","CH4") = Sum((fuel,time, mperiod),p_data(fuel, "CH4Pot") *POW.L(fuel, time, mperiod)/10

sumReport("Env","N2O") = Sum((fuel,time, mperiod),p_data(fuel, "N2OPot") *POW.L(fuel, time, mperiod)/10

emisReport("Env","CO2", time, mperiod, fuel) = (p_data(fuel, "CO2Pot") *POW.L(fuel, time, mperiod) );

emisReport("Env","CH4", time, mperiod, fuel) = (p_data(fuel, "CH4Pot") *POW.L(fuel, time, mperiod) / 10

emisReport("Env","N2O", time, mperiod, fuel) = (p_data(fuel, "N2OPot") *POW.L(fuel, time, mperiod) /100



costReport("Env") = TotCost.L;

display  report, sumReport, emisReport;

execute_unload "pp_time_out.gdx" report, sumReport, costReport, emisReport;
execute 'gdxxrw pp_time_out.gdx output = pp_writingSet.xlsx par = report rng = power!';
execute 'gdxxrw pp_time_out.gdx output = pp_writingSet.xlsx par = sumReport rng = emis!';
execute 'gdxxrw pp_time_out.gdx output = pp_writingSet.xlsx par = costReport rng = cost!';
execute 'gdxxrw pp_time_out.gdx output = pp_writingSet.xlsx par = emisReport rng = emis_day!';
```